

Алгоритмы сортировки
Лекция 2

1.1 Пирамидальная сортировка

Пирамидальная сортировка (heap sort (heap – куча)) – алгоритм сортировки, требующий при сортировке n элементов в худшем, в среднем и в лучшем случае O(n log n) операций. Количество применяемой служебной памяти не зависит от размера массива (то есть, O(1)).

Пирамидальная сортировка сильно **улучшает базовый алгоритм (сортировку выбором**), используя структуру данных «**куча**» для ускорения нахождения и удаления максимального (минимального) элемента.

С другой стороны, пирамидальная сортировка может рассматриваться как **усовершенствованная Bubblesort**, в которой элемент всплывает (max-heap) / тонет (min-heap) по многим путям.

1.2 Пирамидальная сортировка

В компьютерных науках **куча** – это специализированная **структура данных типа дерево**, которая удовлетворяет свойству (кучи):

если узел B является узлом-потомком узла A, то ключ(A) ≥ ключ(B). Из этого следует, что элемент с наибольшим ключом всегда является корневым узлом кучи, поэтому иногда такие кучи называют **max-кучами.**
В альтернативном случае, если сравнение перевернуть, то наименьший элемент будет всегда корневым узлом. Такие кучи называют **min-кучами**.

Не существует никаких ограничений относительно того, сколько узлов-потомков имеет каждый узел кучи, хотя на практике их число обычно не более двух.

Куча является максимально **эффективной реализацией** абстрактного типа данных, который называется **очередью с приоритетом**.

Структуру данных **куча** не следует путать с понятием куча в динамическом распределении памяти.

1.3 Пирамидальная сортировка

Сортировка пирамидой использует ***сортирующее дерево***, которое называется **пирамидой** и является частным случаем кучи.

Пирамида — это полная бинарная куча, то есть для нее выполнены условия:

- Каждый лист имеет глубину либо d, либо d − 1, d — максимальная глубина дерева.
- Значение в любой вершине больше (не меньше), чем значения её потомков.

Пирамиды интересны сами по себе и полезны при реализации **очередей с приоритетом**.

Удобная структура данных для пирамиды — такой массив Array, что Array[1] — элемент в корне, а потомками элемента Array[i] являются Array[2i] и Array[2i+1].

Это известное правило расположения полного двоичного дерева в массиве.

1.4 Пирамидальная сортировка







Индекс	1	2	3	4	5	6	7	8	9	10	11	12	13	14	15
Значение	15	14	13	9	10	12	4	3	1	8	6	7	11	2	5

Примеры структур, являющихся пирамидами.

1.5 Пирамидальная сортировка

Алгоритм построения пирамиды

Можно построить пирамиду **снизу вверх**.

1. Вначале **разместим элементы исходного массива в виде двоичного дерева**, согласно известному правилу (по ширине).
2. Затем **сформируем пирамиды из небольших поддеревьев** (не более, чем с тремя узлами) **внизу дерева**. Для этого сравним вершину маленького дерева с каждым из потомков. Если один из потомков больше, он меняется местами с родителем. Если оба потомка больше, **больший потомок меняется местами с родителем.**

Этот шаг повторяется до тех пор, пока все поддеревья, имеющие по 3 узла, не будут преобразованы в пирамиды.

3. Теперь **объединим маленькие пирамиды и создадим более крупные пирамиды**. Сравним новую вершину с каждым из потомков. Если один из потомков больше, поменяем его местами с новой вершиной.

Если одно поддерево изменилось, проверяем, является ли оно все еще пирамидой. Для этого сравниваем его новую вершину с потомками и, если один из потомков больше, меняем его местами с новой вершиной.

Продолжаем **перемещать новую вершину вниз**. В конце концов, либо будет достигнута точка, в которой перемещаемый вниз узел больше обоих своих потомков, либо алгоритм достигнет основания дерева.

4. Продолжим объединение пирамид, образуя пирамиды большего размера до тех пор, пока все элементы не образуют одну большую пирамиду.

1.6 Пирамидальная сортировка



Несортированный список в полном дереве



Поддеревья второго уровня являются пирамидами



Объединение пирамид в пирамиду большего размера

1.7 Пирамидальная сортировка

Перемещение элемента из положения Queue[parent] **вниз по пирамиде**. Если поддеревья ниже Queue[parent] являются пирамидами, то процедура объединяет пирамиды, образуя пирамиду большего размера.

```
type
   QueueEntry = record
      value:String[10];
      priority:Integer;
end;

// Перемещение элемента вниз по пирамиде, пока он не сможет
// переместиться еще глубже.
procedure HeapPushDown(parent : Integer);
var
   child, top_priority : Integer;
   top_value : String;
begin
   top_priority := Queue[parent].priority;
   top_value := Queue[parent].value;
```

1.8 Пирамидальная сортировка

```
    repeat          // Бесконечный цикл.
        Child := 2 * parent;
        if (child > NumItems) then
            break
        else begin
            // Формируем дочерний узел узлом с большим приоритетом.
            if (child < NumItems) then
                if (Queue[child + 1].priority > Queue[child].priority)
                then
                    child := child + 1;
                if (Queue[child].priority > top_priority) then
                begin
                    // Дочерний узел имеет больший приоритет.
                    // Меняем местами родительский и дочерний узлы.
                    Queue[parent] := Queue[child];
                    // Перемещаем данный дочерний узел вверх.
                    parent := child;
                end else
                    // Родительский узел имеет больший приоритет. Готово.
                    Break;
            end;
        until (False);
        Queue[parent].priority := top_priority;
        Queue[parent].value := top_value;
end;
```

1.9 Пирамидальная сортировка

Полный **алгоритм**, использующий процедуру HeapPushDown **для создания пирамиды из дерева элементов**.

```
procedure BuildHeap;
var
   I : Integer;
begin
   for i := (max + min) div 2 downto min do
      HeapPushDown(i);
end;
```

1.10 Пирамидальная сортировка

Очереди с приоритетом



Очередь с приоритетами на основе связанного списка

Удаление элемента из очереди с приоритетом

Если в качестве очереди с приоритетом используется **пирамида**, легко найти **элемент с самым высоким приоритетом** — он всегда **находится на вершине пирамиды**. Но если его удалить, получившееся дерево без корня уже не будет пирамидой.

Чтобы снова превратить дерево без корня в пирамиду, **поместим последний элемент** (самый правый элемент на нижнем уровне) **в вершину пирамиды**. Затем при помощи процедуры HeapPushDown **продвинем новый корневой узел вниз по дереву** до тех пор, пока дерево снова не станет пирамидой. В этот момент, можно получить **на выходе очереди с приоритетом следующий элемент с наивысшим приоритетом.**

1.11 Пирамидальная сортировка

Добавление элемента к очереди с приоритетом

Поместим новый элемент на свободное место в конце массива. Полученное дерево может не быть пирамидой.

Чтобы снова преобразовать его в пирамиду, сравним новый элемент с его родителем. **Если новый элемент больше родителя, поменяем их местами**. Если элемент больше родителя, то он также больше и второго потомка.

Продолжим сравнение нового элемента с родителем и перемещение его по дереву вверх к корню, пока не найдется родитель, больший, чем новый элемент (или пока не достигнем корня). В этот момент, дерево снова представляет собой пирамиду.

1.12 Пирамидальная сортировка

Анализ пирамид

При первоначальном превращении списка в пирамиду осуществляется создание множества пирамид меньшего размера. **Для каждого внутреннего узла дерева строится пирамида с корнем в этом узле**. Если дерево содержит N элементов, то в дереве O(N) внутренних узлов, и в итоге **приходится создать O(N) пирамид**.

При создании каждой пирамиды может потребоваться продвигать элемент вниз по пирамиде, возможно до тех пор, пока он не достигнет концевого узла. **Самые высокие из построенных пирамид будут иметь высоту порядка O(log(N)).** Так как создается O(N) пирамид, и для построения самой высокой из них требуется O(log(n)) шагов, то **все пирамиды можно построить за время порядка O(N * log(N)).**

На самом деле времени потребуется еще меньше – порядка O(N).

1.13 Пирамидальная сортировка

Время для построения пирамиды

Пусть H — высота дерева. Это полное двоичное дерево, следовательно, H=log(N).

Для каждого узла, который находится на расстоянии H-i уровней от корня дерева, строится пирамида с высотой i. Всего таких узлов 2^(H-i), и всего **создается 2^(H-i) пирамид с высотой i**.

Перемещение элемента вниз по пирамиде с высотой i требует до i шагов. **Для пирамид с высотой i полное число шагов**, которое потребуется для построения 2^(H-i) пирамид, равно **i*2^(H-i)**.

Сложив все шаги, затрачиваемые на построение пирамид разного размера, получаем 1*2^(H-1)+2*2^(H-2)+3*2^(H-3) +…+(H-1)* 2^1. Вынеся за скобки 2^H, получим 2^H*(1/2+2/2^2+3/2^3+…+(H-1)/2^(H-1))<2^H*2=**N*2**.

Это означает, что **для первоначального построения пирамиды требуется порядка O(N) шагов.**

1.14 Пирамидальная сортировка

Время для удаления максимального элемента

Для удаления элемента из очереди с приоритетом, **последний элемент перемещается на вершину дерева.** Затем **продвигается вниз**, пока не займет свое окончательное положение, и дерево снова не станет пирамидой. Так как дерево имеет высоту log(N), процесс может занять не более log(N) шагов. Это означает, что удаление **элемента из очереди с приоритетом на основе пирамиды осуществляется за O(log(N)) шагов.**

Время добавления элемента к очереди с приоритетом

При добавлении в пирамиду **новый элемент помещается внизу дерева и передвигается к вершине**, пока не займет нужное место (максимум за log(N) шагов). То есть, **новый элемент добавляется к очереди с приоритетом на основе пирамиды тоже за время порядка O(log(N)) .**

1.15 Пирамидальная сортировка

Алгоритм пирамидальной сортировки

Алгоритм пирамидальной сортировки использует уже описанные алгоритмы для работы с пирамидами. **Идея состоит в том, чтобы создать очередь с приоритетом и последовательно удалять по одному элементу из очереди.**

Для удаления элемента алгоритм меняет его местами с последним элементом в пирамиде. Он помещает удаленный элемент в конец массива и **уменьшает счетчик элементов списка**, чтобы исключить из рассмотрения последнюю позицию.

Новый элемент на вершине может оказаться меньше, чем его потомки. Поэтому **алгоритм продвигает новый элемент вниз на его место**, используя процедуру HeapPushDown.

Алгоритм **продолжает менять элементы местами и восстанавливать пирамиду до тех пор, пока в пирамиде не останется элементов.**

1.16 Пирамидальная сортировка

Алгоритм сортировки состоит из двух основных шагов:

1.Выстраиваем элементы массива в виде сортирующего дерева:

Array[i] >= Array[2i]

Array[i] >= Array[2i+1]

при 1<=i<n/2.

Этот шаг требует O(n) операций.

2.Будем удалять элементы из корня по одному за раз и перестраивать дерево. То есть на первом шаге обмениваем Array[1] и Array[n], преобразовываем Array[1], Array[2], … , Array[n-1] в сортирующее дерево.

Затем переставляем Array[1] и Array[n-1], преобразовываем Array[1], Array[2], … , Array[n-2] в сортирующее дерево.

Процесс продолжается до тех пор, пока в сортирующем дереве не останется один элемент. Тогда Array[1], Array[2], … , Array[n] — упорядоченная последовательность.

1.17 Пирамидальная сортировка

Время выполнения алгоритма пирамидальной сортировки

Первоначальное построение пирамиды требует **O(N)** шагов.

После этого требуется **O(log(N)) шагов для восстановления пирамиды, когда один элемент продвигается вниз на свое место**.

Это **действие выполняется N раз**, поэтому **требуется всего порядка** O(N)*O(log(N))=**O(N*log(N)) шагов, чтобы получить из пирамиды упорядоченный список.**

Полное время выполнения для алгоритма пирамидальной сортировки составляет порядка O(N)+O(N*log(N))=**O(N*log(N)).**

1.18 Пирамидальная сортировка

ПРИМЕР ПИРАМИДАЛЬНОЙ СОРТИРОВКИ

K_1	K_2	K_3	K_4	K_5	K_6	K_7	K_8	K_9	K_{10}	K_{11}	K_{12}	K_{13}	K_{14}	K_{15}	K_{16}
503	087	512	061	908	170	897	275	[653	426	154	509	612	677	765	703]
503	087	512	061	908	170	897	[703	653	426	154	509	612	677	765	275]
503	087	512	061	908	170	[897	703	653	426	154	509	612	677	765	275]
503	087	512	061	908	[612	897	703	653	426	154	509	170	677	765	275]
503	087	512	061	[908	612	897	703	653	426	154	509	170	677	765	275]
503	087	512	[703	908	612	897	275	653	426	154	509	170	677	765	061]
503	087	[897	703	908	612	765	275	653	426	154	509	170	677	512	061]
503	[908	897	703	426	612	765	275	653	087	154	509	170	677	512	061]
[908	703	897	653	426	612	765	275	503	087	154	509	170	677	512	061]
[897	703	765	653	426	612	677	275	503	087	154	509	170	061	512]	908
[765	703	677	653	426	612	512	275	503	087	154	509	170	061]	897	908
[703	653	677	503	426	612	512	275	061	087	154	509	170]	765	897	908
[677	653	612	503	426	509	512	275	061	087	154	170]	703	765	897	908
[653	503	612	275	426	509	512	170	061	087	154]	677	703	765	897	908
[612	503	512	275	426	509	154	170	061	087]	653	677	703	765	897	908
[512	503	509	275	426	087	154	170	061]	612	653	677	703	765	897	908
[509	503	154	275	426	087	061	170]	512	612	653	677	703	765	897	908
[503	426	154	275	170	087	061]	509	512	612	653	677	703	765	897	908
[426	275	154	061	170	087]	503	509	512	612	653	677	703	765	897	908
[275	170	154	061	087]	426	503	509	512	612	653	677	703	765	897	908
[170	087	154	061]	275	426	503	509	512	612	653	677	703	765	897	908
[154	087	061]	170	275	426	503	509	512	612	653	677	703	765	897	908
[087	061]	154	170	275	426	503	509	512	612	653	677	703	765	897	908

1.19 Пирамидальная сортировка

Пример. Пирамидальная сортировка (в узлах указаны номера элементов массива).



Последовательное распределение памяти для полного бинарного дерева.

1.20 Пирамидальная сортировка

```
procedure TSortForm.Heapsort(list : PLongintArray;
                             min, max : Longint);
var
   i, tmp : Longint;
begin
   // Построение пирамиды (за исключением корневого узла).
   for i := (max + min) div 2 downto min + 1 do
      HeapPushDown(list,i,max);

   // Повторить:
   // 1. HeapPushDown.
   // 2. Вывод корневого узла.
   for i := max downto min + 1 do
   begin
      // 1. HeapPushDown.
      HeapPushDown(list,min,i);

      // 2. Вывод корневого узла.
      tmp := list^[min];
      list^[min] := list^[i];
      list^[i] := tmp;
   end;
end;
```

1.21 Пирамидальная сортировка

Достоинства и недостатки алгоритма пирамидальной сортировки

Достоинства

- Имеет доказанную **оценку худшего случая O(n log n).**
- **Не зависит от значений или распределения элементов** до начала сортировки (как и сортировка слиянием).
- **Хорошо работает со списками, содержащими большое число одинаковых элементов** (в отличие от быстрой сортировки).
- **Не требует дополнительного пространства** для хранения временных значений, создает первоначальную пирамиду и упорядочивает элементы в пределах исходного массива списка. Пригодна для больших списков.

Недостатки

- Сложен в реализации.
- Неустойчив — для обеспечения устойчивости нужно расширять ключ.
- Не обладает свойством естественности: на почти отсортированных массивах работает столь же долго, как и на хаотических данных.
- Работает немного медленнее, чем сортировка слиянием.

2.1 Сортировка подсчетом

Сортировка подсчетом (counting sort) — специализированный алгоритм, который очень хорошо работает, если **элементы данных — целые числа, значения которых находятся в относительно узком диапазоне**, например, если значения находятся между 1 и 1000.

Выдающаяся скорость сортировки подсчетом, значительно быстрее быстрой сортировки, достигается за счет того, что при этом **не используются операции сравнения элементов**. Время выполнения любого алгоритма сортировки, использующего операции сравнения, порядка O(N*log(N)). **Без использования операций сравнения элементов, алгоритм сортировки подсчетом позволяет упорядочивать элементы за время порядка O(N).**

2.2 Сортировка подсчетом

Алгоритм сортировки подсчетом

1 шаг. **Создается массив для подсчета числа элементов, имеющих определенное значение**. Если значения элементов сортируемого массива List находятся в диапазоне между min_value и max_value, алгоритм создает массив Counts с нижней границей индекса min_value и верхней границей индекса max_value. **Если существует M значений элементов,** массив Counts содержит M записей, и **время выполнения этого шага порядка O(M).**

2 шаг. **Вычисляется, сколько раз в списке встречается каждое значение.** Для каждого значения i между min_value и max_value сортируемого массива, в массиве Counts увеличивается значение соответствующей записи Counts(List(i)). Так как **этот шаг просматривает все записи в исходном массиве, он требует порядка O(N) шагов.**

3 шаг. Обходится массив счетчиков и **помещается соответствующее число элементов в отсортированный массив**. Для каждого значения i между min_value и max_value, в массив помещается Counts(i) элементов со значением i. Так как **этот шаг** помещает по одной записи в каждую позицию в **отсортированном** массиве, он **требует порядка O(N) шагов**.

2.3 Сортировка подсчетом

Время работы алгоритма

Алгоритм целиком **требует порядка** O(M)+O(N)+O(N)=**O(M+N) шагов**.

Если M<<N, то **O(M+N)=O(N),** что довольно быстро.

Пример. Если N=100000 и M=1000, то M+N=101000, тогда как N*log(N)=1,6 миллиона (для алгоритмов, использующих сравнения).

Шаги, выполняемые алгоритмом сортировки подсчетом, также **относительно просты** по сравнению с шагами быстрой сортировки.

Сортировка подсчетом опирается на тот факт, что значения данных — целые числа, поэтому этот **алгоритм не может сортировать данные других типов.**

2.4 Сортировка подсчетом

```
type
   TCountArray = array [1..10000000] of Longint;
   PCountArray = ^TCountArray;

procedure Countingsort(list : PLongintArray; counts : PCountArray;
                       min, max, min_value, max_value : Longint);
var
   i, j, new_index : Longint;
```

2.5 Сортировка подсчетом

```
begin
    // Установка счетчиков в 0.
    for i := min_value to max_value do
        counts^[i] := 0;

    // Подсчет значений.
    for i := min to max do
        counts^[list^[i]] := counts^[list^[i]] + 1;

    // Помещение значений в правильную позицию.
    new_index := min;
    for i := min_value to max_value do
        for j := 1 to counts^[i] do
        begin
            list^[new_index] := i;
            new_index := new_index + 1;
        end;
end;
```

3.1 Блочная сортировка

Блочная сортировка (карманная сортировка, корзинная сортировка, англ. Bucket sort) — алгоритм сортировки, в котором **сортируемые элементы распределяются между конечным числом отдельных блоков** (карманов, корзин) **так, чтобы все элементы в каждом следующем по порядку блоке были всегда больше** (или меньше), **чем в предыдущем**. Каждый блок затем сортируется отдельно, либо рекурсивно тем же методом, либо другим. Затем элементы помещаются обратно в массив.

Особенности алгоритма

Этот тип сортировки может обладать **линейным временем исполнения.**

Не использует операций **сравнения элементов** (в основной части алгоритма).

Данный алгоритм **требует знаний о природе сортируемых данных**, выходящих за рамки функций "сравнить" и "поменять местами", достаточных для сортировки слиянием, сортировки пирамидой, быстрой сортировки, сортировки Шелла, сортировки вставкой. Например, **знание значений максимального и минимального элементов**.

3.2 Блочная сортировка

Алгоритм

- **Алгоритм использует значения элементов для разбиения их на на множество блоков,** и затем последовательно **рекурсивно сортирует полученные блоки**.
- **Когда блоки становятся достаточно малыми, алгоритм останавливается** и использует более простой алгоритм типа сортировки выбором для завершения процесса.
- **Отсортированный массив получается путем последовательного перечисления элементов каждого блока.**

Предположения

Для деления массива на блоки, алгоритм предполагает, что **значения данных распределены равномерно**, и распределяет элементы по блокам тоже равномерно.

Поскольку входные числа распределены равномерно, предполагается, что в каждый блок попадет приблизительно одинаковое количество чисел.

Непрерывное равномерное распределение — распределение случайной вещественной величины, принимающей значения, принадлежащие интервалу [a, b], характеризующееся тем, что плотность вероятности на этом интервале постоянна.

Случайная величина имеет **дискретное равномерное распределение**, если она принимает конечное число значений с равными вероятностями.

3.3 Блочная сортировка

Сложность алгоритма

Если в списке N элементов, и алгоритм использует N блоков, в каждый блок попадает всего один или два элемента. Программа может отсортировать их за конечное число шагов, поэтому **время выполнения алгоритма в целом порядка O(N).**

Преимущества алгоритма

Относится к классу быстрых алгоритмов с **линейным временем исполнения O(N)** (на удачных входных данных).

Недостатки алгоритма

Сильно **деградирует при** большом количестве мало отличных (**одинаковых) элементов**, или на неудачной функции получения номера блока по содержимому элемента.

Проблемы могут возникать, если список содержит небольшое число различных значений.

Например, если все элементы имеют одно и то же значение, они все будут помещены в один блок. Если алгоритм не обнаружит это, он снова и снова будет помещать все элементы в один и тот же блок, вызвав бесконечную рекурсию и исчерпав все стековое пространство.

3.4 Блочная сортировка

Пример. Число блоков (корзин) меньше числа элементов.



Элементы распределяются по корзинам

Затем элементы в каждой корзине сортируются

3.5 Блочная сортировка

Пример. Число блоков равно числу элементов

Помещение элементов в блоки

3.6 Блочная сортировка

Реализовать **алгоритм блочной сортировки** можно различными способами.

Блочная сортировка на основе одномерного массива

Блочную сортировку можно реализовать в массиве, используя идеи подобные тем, которые используются при сортировке подсчетом.

При каждом вызове алгоритма, **вначале подсчитывается число элементов, которые относятся к каждому блоку**.

Потом на основе этих данных **рассчитываются смещения во вре'менном массиве**, которые затем используются для правильного расположения элементов в массиве.

В конце концов, **блоки рекурсивно сортируются**, и **отсортированные данные перемещаются** обратно **в исходный массив**.

3.6 Блочная сортировка

Блочная сортировка с использованием связанных списков

Можно использовать в качестве блоков связанные списки. Это облегчает перемещение элементов из одного блока в другой в процессе работы алгоритма.

Этот метод может быть более сложным, если элементы изначально расположены в массиве. В этом случае, необходимо перемещать элементы из массива в связанный список и обратно в массив после завершения сортировки. **Для создания связанного списка также требуется дополнительная память**.

3.7 Блочная сортировка

```
type
   PCell = ^TCell;
   TCell = record
      Value : Longint;          // Данные.
      NextCell : PCell;         // Следующая ячейка.
   end;

   TCellArray = array [1..1000000] of TCell;
   PCellArray = ^TCellArray;

procedure LLBucketsort(top : PCell);
var
   count, min_value, max_value : Longint;
   i, value, bucket_num : Longint;
   cell, nxt : PCell;
   bucket : PCellArray;
   scale : Double;
begin
   cell := top^.NextCell;
   if (cell = nil) then exit;
```

3.8 Блочная сортировка

```
// Подсчет ячеек и поиск минимального и максимального значений.
count := 1;
min_value := cell^.Value;
max_value := min_value;
cell := cell^.NextCell;
while (cell<>nil) do
begin
    count := count + 1;
    value := cell^.value;
    if (min_value > value) then min_value := value;
    if (max_value < value) then max_value := value;
    cell := cell^.NextCell;
end;

// Если min_value = max_value, то имеется только одно значение,
// поэтому список отсортирован.
if (min_value = max_value) then exit;

// Если список содержит не более Cutoff ячеек, заканчиваем
// сортировку с помощью LLInsertionsort.
if (count <= Cutoff) then
begin
    LLInsertionsort(top);
    exit;
end;
```

3.9 Блочная сортировка

```
// Размещение пустых блоков.
GetMem(bucket,count * SizeOf(TCell));
for i := 1 to count do bucket^[i].NextCell := nil;

// Перемещение ячеек в блоки.
Scale := (count - 1) / (max_value - min_value);
Cell := top^.NextCell;
while (cell<>nil) do
begin
    nxt := cell^.NextCell;
    value := cell^.value;
    if (value = max_value) then
        bucket_num := count
    else
        bucket_num := Trunc((value - min_value) * scale) + 1;
    cell^.NextCell := bucket^[bucket_num].NextCell;
    bucket^[bucket_num].NextCell := cell;
    cell := nxt;
end;
```

3.10 Блочная сортировка

```
    // Объединение сортированных списков.
    top^.NextCell := bucket^[count].NextCell;
    for i := count - 1 downto 1 do
    begin
        cell := bucket^[i].NextCell;
        if (cell<>nil) then
        begin
            nxt := cell^.NextCell;
            while   nxt<>nil   do
            begin
                cell := nxt;
                nxt := cell^.NextCell;
            end;
            cell^.NextCell := top^.NextCell;
            top^.NextCell := bucket^[i].NextCell;

            // Освобождаем метку конца, если она есть.
            if (nxt<>nil) then Dispose(nxt);
        end;
    end;

    // Освобождаем память, выделенную для блоков.
    FreeMem(bucket);
end;
```

4.1 Распределяющая сортировка

Распределяющая сортировка относится **к быстрым алгоритмам, не использующим операции сравнения**, с **временем выполнения порядка O(N)** и является разновидностью блочной сортировки.

Предположим, что **элементы** линейного списка B есть **T-разрядные положительные десятичные числа**.

D(j, n) — j-я справа цифра в десятичной записи числа n>=0,

т.е. D(j, n)= trunc (n/m) mod 10, где m=10^(j-1),

или D(j, n)= floor (n/m)%10, где m=10^(j-1) .

Пусть B0,B1,...,B9 — вспомогательные списки (карманы), вначале пустые.

Алгоритм

Для реализации распределяющей сортировки для каждого j=1,2,...,T выполняется процедура, состоящая из двух процессов, называемых **распределение и сборка.**

Распределение заключается в том, что **элемент Ki** (i=1,…,N) из B добавляется как последний в список Bm, где m = D(j, Ki), и таким образом **получаются десять списков, в каждом из которых j- тые разряды чисел одинаковы и равны m.**

Сборка объединяет списки B0,B1,...,B9 в этом же порядке, образуя **один список B**.

4.2 Распределяющая сортировка

Пример

Рассмотрим реализацию распределяющей сортировки при T=2 для списка:

B=<09, 07, 18, 03, 52, 04, 06, 08, 05, 13, 42, 30, 35, 26> .

J=1.

Распределение 1:

B0=<30>, B1=<>, B2=<52, 42>, B3=<03, 13>, B4=<04>,
B5=<05, 35>, B6=<06, 26>, B7=<07>, B8=<18, 08>, B9=<09>.

Сборка 1:

B=<30, 52, 42, 03, 13, 04, 05, 35, 06, 26, 07, 18, 08, 09>

J=2.

Распределение 2:

B0=<03, 04, 05, 06, 07, 08, 09>, B1=<13, 18>, B2=<26>,
B3=<30, 35>, B4=<42>, B5=<52>, B6=<>, B7=<>, B8=<>, B9=<>.

Сборка 2:

B=<03, 04, 05, 06, 07, 08, 09, 13, 18, 26, 30, 35, 42, 52>.

4.3 Распределяющая сортировка

Время выполнения

Количество действий, необходимых для сортировки списка из N T-значных чисел, определяется как O(N*T). Если T<<N, то **время выполнения** алгоритма распределяющей сортировки **порядка O(N).**

Недостатки

Сортирует только целые числа.

Требует использования дополнительной памяти под карманы.

Однако можно исходный список представить как связанный и сортировку организовать так, чтобы для карманов B0, B1,...,B9 не использовать дополнительной памяти, элементы списка не перемещать, а с помощью перестановки указателей присоединять их к тому или иному карману.

4.4 Распределяющая сортировка

Разновидности распределяющей сортировки

Битовая сортировка

Элементы списка интерпретируются как двоичные числа, и D(j, n) обозначает j-ю справа двоичную цифру числа n.

В процессе **распределения** требуются только два вспомогательных кармана: B0 и B1; их можно разместить в одном массиве, двигая списки B0 и B1 навстречу друг другу и отмечая точку встречи.

Для осуществления **сборки** нужно за списком B0 написать инвертированный список B1.

Выделение j-го бита требует только операций сдвига.

Бинарная сортировка

Из всех элементов списка B выделяются его минимальный и максимальный элементы и находится их среднее арифметическое m =(MIN+MAX)/2.

Список B разбивается на подсписки B1 и B2, причем **в B1 попадают элементы, не большие m, а в список B2 - элементы, большие m.**

Для непустых подсписков B1 и B2 сортировка продолжается **рекурсивно**.